

```
/**
  In this example, we do a little bit more involved image processing.

  The goal is a image that mirrors the left side across the center
  line onto the right side. There are two equations that govern this. So you
  can see the difference, this displays the original image and then the mirrored
  image below it.

  First, we need to figure out what the mirrored point is. If the point we
  want to mirror is at (i,j), then the mirrored point is (img.width - i - 1,j).

  Second, we need to figure out where these coordinates can be found in our 1D
  array. Given a point(i,j), the index is i + j*img.width.

  C. Andrews
  2014-01-16
  **/
```

```
PImage img;

void setup() {
  img = loadImage("cole2.jpg");
  size(img.width, img.height*2);
  noLoop();
}

void draw() {
  // display the original image
  image(img, 0, 0);

  // we need to load the pixels so we can access them in the pixels array
  img.loadPixels();

  // loop over half of the pixels in the image
  for (int j = 0; j < img.height; j++) {
    for (int i = 0; i < img.width/2; i++) {
      int srcIndex, dstIndex;
      // convert i,j into an index for the source pixel
      srcIndex = i + j * img.width;
      // convert i,j into an index for the destination pixel
      dstIndex = (img.width - i - 1) + j * img.width;

      // copy the value from the source image to the destination location
      img.pixels[dstIndex] = img.pixels[srcIndex];
    }
  }

  // update the pixels to put them back in the image
  img.updatePixels();

  // display the image
  image(img, 0, img.height);
}
```